The Sophomore's Dream

Manuel Eberl

March 19, 2025

Abstract

This article provides a brief formalisation of the two equations known as the *Sophomore's Dream*, first discovered by Johann Bernoulli [1] in 1697:

$$\int_0^1 x^{-x}\,\mathrm{d}x = \sum_{n=1}^{\infty} n^{-n} \quad \text{and} \quad \int_0^1 x^{x}\,\mathrm{d}x = -\sum_{n=1}^{\infty} (-n)^{-n}$$

Contents

1 The Sophomore's Dream

theory *Sophomores-Dream*
  imports *HOL−Analysis.Analysis HOL−Real-Asymp.Real-Asymp*
begin

This formalisation mostly follows the very clear proof sketch from Wikipedia [3]. That article also provides an interesting historical perspective. A more detailed exploration of Bernoulli's historical proof can be found in the book by Dunham [2].

The name 'Sophomore's Dream' apparently comes from a book by Borwein et al., in analogy to the 'Freshman's Dream' equation $(x + y)^n = x^n + y^n$ (which is generally *not* true except in rings of characteristic n).

1.1 Continuity and bounds for $x \log x$

lemma *x-log-x-continuous*: *continuous-on* $\{0..1\}$ $(\lambda x{::}real.\ x * ln\ x)$
proof −
  have *continuous* (*at x within* $\{0..1\}$) $(\lambda x{::}real.\ x * ln\ x)$ **if** $x \in \{0..1\}$ **for** x

```
  proof (cases x = 0)
    case True
    have ((λx::real. x * ln x) ⟶ 0) (at-right 0)
      by real-asymp
    thus ?thesis using True
      by (simp add: continuous-def Lim-ident-at at-within-Icc-at-right)
  qed (auto intro!: continuous-intros)
  thus ?thesis
    using continuous-on-eq-continuous-within by blast
qed

lemma x-log-x-within-01-le:
  assumes x ∈ {0..(1::real)}
  shows   x * ln x ∈ {−exp (−1)..0}
proof −
  have x * ln x ≤ 0
    using assms by (cases x = 0) (auto simp: mult-nonneg-nonpos)
  let ?f = λx::real. x * ln x
  have diff: (?f has-field-derivative (ln x + 1)) (at x) if x > 0 for x
    using that by (auto intro!: derivative-eq-intros)
  have diff′: ?f differentiable at x if x > 0 for x
    using diff[OF that] real-differentiable-def by blast

  consider x = 0 | x = 1 | x = exp (−1) | 0 < x x < exp (−1) | exp (−1) < x x
< 1
    using assms unfolding atLeastAtMost-iff by linarith
  hence x * ln x ≥ −exp (−1)
  proof cases
    assume x: 0 < x x < exp (−1)
    have ∃l z. x < z ∧ z < exp (−1) ∧ (?f has-real-derivative l) (at z) ∧
            ?f (exp (−1)) − ?f x = (exp (−1) − x) * l
      using x by (intro MVT continuous-on-subset [OF x-log-x-continuous] diff′)
auto
    then obtain l z where lz:
      x < z z < exp (−1) (?f has-real-derivative l) (at z)
      ?f x = −exp (−1) − (exp (−1) − x) * l
      by (auto simp: algebra-simps)
    have [simp]: l = ln z + 1
      using DERIV-unique[OF diff[of z] lz(3)] lz(1) x by auto
    have ln z ≤ ln (exp (−1))
      using lz x by (subst ln-le-cancel-iff) auto
    hence (exp (− 1) − x) * l ≤ 0
      using x lz by (intro mult-nonneg-nonpos) auto
    with lz show ?thesis
      by linarith
  next
    assume x: exp (−1) < x x < 1
    have ∃l z. exp (−1) < z ∧ z < x ∧ (?f has-real-derivative l) (at z) ∧
            ?f x − ?f (exp (−1)) = (x − exp (−1)) * l
```

```
    proof (intro MVT continuous-on-subset [OF x-log-x-continuous] diff')
      fix t :: real assume t: exp (−1) < t
      show t > 0
        by (rule less-trans [OF - t]) auto
    qed (use x in auto)
    then obtain l z where lz:
      exp (−1) < z z < x (?f has-real-derivative l) (at z)
      ?f x = −exp (−1) − (exp (−1) − x) ∗ l
      by (auto simp: algebra-simps)
    have z > 0
      by (rule less-trans [OF - lz(1)]) auto
    have [simp]: l = ln z + 1
      using DERIV-unique[OF diff[of z] lz(3)] ‹z > 0› by auto
    have ln z ≥ ln (exp (−1))
      using lz ‹z > 0› by (subst ln-le-cancel-iff) auto
    hence (exp (− 1) − x) ∗ l ≤ 0
      using x lz by (intro mult-nonpos-nonneg) auto
    with lz show ?thesis
      by linarith
  qed auto

  with ‹x ∗ ln x ≤ 0› show ?thesis
    by auto
qed
```

1.2 Convergence, Summability, Integrability

As a first result we can show that the two sums that occur in the two different versions of the Sophomore's Dream are absolutely summable. This is achieved by a simple comparison test with the series $\sum_{k=1}^{\infty} k^{-2}$, as $k^{-k} \in O(k^{-2})$.

```
theorem abs-summable-sophomores-dream: summable (λk. 1 / real (k ^ k))
proof (rule summable-comparison-test-bigo)
  show (λk. 1 / real (k ^ k)) ∈ O(λk. 1 / real k ^ 2)
    by real-asymp
  show summable (λn. norm (1 / real n ^ 2))
   using inverse-power-summable[of 2, where ?'a = real] by (simp add: field-simps)
qed
```

The existence of the integral is also fairly easy to show since the integrand is continuous and the integration domain is compact. There is, however, one hiccup: The integrand is not actually continuous.

We have $\lim_{x\to 0} x^x = 1$, but in Isabelle 0^0 is defined as *0* (for real numbers). Thus, there is a discontinuity at $x = 0$

However, this is a removable discontinuity since for any $x > 0$ we have $x^x = e^{x \log x}$, and as we have just shown, $e^{x \log x}$ *is* continuous on $[0, 1]$. Since the two integrands differ only for $x = 0$ (which is negligible), the integral still exists.

theorem *integrable-sophomores-dream*: (λx::*real*. *x powr x*) *integrable-on* {*0..1*}
proof −
  have (λx::*real*. *exp* (*x* ∗ *ln x*)) *integrable-on* {*0..1*}
    by (*intro integrable-continuous-real continuous-on-exp x-log-x-continuous*)
  also have *?this* ⟷ (λx::*real*. *exp* (*x* ∗ *ln x*)) *integrable-on* {*0*<..<*1*}
    by (*simp add*: *integrable-on-Icc-iff-Ioo*)
  also have . . . ⟷ (λx::*real*. *x powr x*) *integrable-on* {*0*<..<*1*}
    by (*intro integrable-cong*) (*auto simp*: *powr-def*)
  also have . . . ⟷ *?thesis*
    by (*simp add*: *integrable-on-Icc-iff-Ioo*)
  finally show *?thesis* **.**
qed

Next, we have to show the absolute convergence of the two auxiliary sums that will occur in our proofs so that we can exchange the order of integration and summation. This is done with a straightforward application of the Weierstraß M test.

lemma *uniform-limit-sophomores-dream1*:
  uniform-limit {*0*..(*1*::*real*)}
    (λn *x*. $\sum$*k*<*n*. (*x* ∗ *ln x*) ^ *k* / *fact k*)
    (λx. $\sum$*k*. (*x* ∗ *ln x*) ^ *k* / *fact k*)
    sequentially
proof (*rule Weierstrass-m-test*)
  show *summable* (λk. *exp* (−*1*) ^ *k* / *fact k* :: *real*)
    using *summable-exp*[*of exp* (−*1*)] **by** (*simp add*: *field-simps*)
next
  fix *k* :: *nat* **and** *x* :: *real*
  assume *x*: *x* ∈ {*0..1*}
  have *norm* ((*x* ∗ *ln x*) ^ *k* / *fact k*) = *norm* (*x* ∗ *ln x*) ^ *k* / *fact k*
    by (*simp add*: *power-abs*)
  also have . . . ≤ *exp* (−*1*) ^ *k* / *fact k*
    by (*intro divide-right-mono power-mono*) (*use x-log-x-within-01-le* [*of x*] *x* **in** *auto*)
  finally show *norm* ((*x* ∗ *ln x*) ^ *k* / *fact k*) ≤ *exp* (− *1*) ^ *k* / *fact k* **.**
qed

lemma *uniform-limit-sophomores-dream2*:
  uniform-limit {*0*..(*1*::*real*)}
    (λn *x*. $\sum$*k*<*n*. (−(*x* ∗ *ln x*)) ^ *k* / *fact k*)
    (λx. $\sum$*k*. (−(*x* ∗ *ln x*)) ^ *k* / *fact k*)
    sequentially
proof (*rule Weierstrass-m-test*)
  show *summable* (λk. *exp* (−*1*) ^ *k* / *fact k* :: *real*)
    using *summable-exp*[*of exp* (−*1*)] **by** (*simp add*: *field-simps*)
next
  fix *k* :: *nat* **and** *x* :: *real*
  assume *x*: *x* ∈ {*0..1*}
  have *norm* ((−*x* ∗ *ln x*) ^ *k* / *fact k*) = *norm* (*x* ∗ *ln x*) ^ *k* / *fact k*

by (*simp add*: *power-abs*)
also have ... ≤ *exp* (−*1*) ^ *k* / *fact k*
by (*intro divide-right-mono power-mono*) (*use x-log-x-within-01-le* [*of x*] *x* **in** *auto*)
finally show *norm* ((−(*x* ∗ *ln x*)) ^ *k* / *fact k*) ≤ *exp* (− *1*) ^ *k* / *fact k* **by** *simp*
qed

1.3 An auxiliary integral

Next we compute the integral

$$\int_0^1 (x \log x)^n \,\mathrm{d}x = \frac{(-1)^n\, n!}{(n+1)^{n+1}}\ ,$$

which is a key ingredient in our proof.

lemma *sophomores-dream-aux-integral*:
((λ*x*. (*x* ∗ *ln x*) ^ *n*) *has-integral* (− *1*) ^ *n* ∗ *fact n* / *real* ((*n* + *1*) ^ (*n* + *1*))) {*0*<..<*1*}
proof −
have ((λ*t*. *t powr real n* / *exp t*) *has-integral fact n*) {*0*..}
using *Gamma-integral-real*[*of n* + *1*] **by** (*auto simp*: *Gamma-fact powr-realpow*)
also have *?this* ⟷ ((λ*t*. *t powr real n* / *exp t*) *has-integral fact n*) {*0*<..}
proof (*rule has-integral-spike-set-eq*)
have *eq*: {*x* ∈ {*0*<..} − {*0*..}. *x powr real n* / *exp x* ≠ *0*} = {}
by *auto*
thus *negligible* {*x* ∈ {*0*<..} − {*0*..}. *x powr real n* / *exp x* ≠ *0*}
by (*subst eq*) *auto*
have {*x* ∈ {*0*..} − {*0*<..}. *x powr real n* / *exp x* ≠ *0*} ⊆ {*0*}
by *auto*
moreover have *negligible* {*0*::*real*}
by *simp*
ultimately show *negligible* {*x* ∈ {*0*..} − {*0*<..}. *x powr real n* / *exp x* ≠ *0*}
by (*meson negligible-subset*)
qed
also have ... ⟷ ((λ*t*::*real*. *t* ^ *n* / *exp t*) *has-integral fact n*) {*0*<..}
by (*intro has-integral-spike-eq*) (*auto simp*: *powr-realpow*)
finally have *1*: ((λ*t*::*real*. *t* ^ *n* / *exp t*) *has-integral fact n*) {*0*<..} **.**

have (λ*x*::*real*. |*x*| ^ *n* / *exp x*) *integrable-on* {*0*<..} ⟷
(λ*x*::*real*. *x* ^ *n* / *exp x*) *integrable-on* {*0*<..}
by (*intro integrable-cong*) *auto*
hence *2*: (λ*t*::*real*. *t* ^ *n* / *exp t*) *absolutely-integrable-on* {*0*<..}
using *1* **by** (*simp add*: *absolutely-integrable-on-def power-abs has-integral-iff*)

define *g* :: *real* ⇒ *real* **where** *g* = (λ*x*. −*ln x* ∗ (*n* + *1*))
define *g′* :: *real* ⇒ *real* **where** *g′* = (λ*x*. −(*n* + *1*) / *x*)
define *h* :: *real* ⇒ *real* **where** *h* = (λ*u*. *exp* (−*u* / (*n* + *1*)))
have *bij*: *bij-betw g* {*0*<..<*1*} {*0*<..}

by (*rule bij-betwI*[*of - - - h*]) (*auto simp*: *g-def h-def mult-neg-pos*)
have *deriv*: (*g has-real-derivative g′ x*) (*at x within* {*0*<..<*1*})
if $x \in$ {*0*<..<*1*} **for** *x*
unfolding *g-def g′-def* **using** *that* **by** (*auto intro*!: *derivative-eq-intros simp*: *field-simps*)

have (λ*t*::*real*. *t* ^ *n* / *exp t*) *absolutely-integrable-on g* ‘ {*0*<..<*1*} ∧
integral (*g* ‘ {*0*<..<*1*}) (λ*t*::*real*. *t* ^ *n* / *exp t*) = *fact n*
using *1 2 bij* **by** (*simp add*: *bij-betw-def has-integral-iff*)
also have *?this* ⟷ ((λ*x*. |*g′ x*| $*_R$ (*g x* ^ *n* / *exp* (*g x*))) *absolutely-integrable-on* {*0*<..<*1*} ∧
integral {*0*<..<*1*} (λ*x*. |*g′ x*| $*_R$ (*g x* ^ *n* / *exp* (*g x*))) = *fact n*)
by (*intro has-absolute-integral-change-of-variables-1′* [*symmetric*] *deriv*)
(*auto simp*: *inj-on-def g-def*)
finally have ((λ*x*. |*g′ x*| $*_R$ (*g x* ^ *n* / *exp* (*g x*))) *has-integral fact n*) {*0*<..<*1*}
using *eq-integralD set-lebesgue-integral-eq-integral*(*1*) **by** *blast*
also have *?this* ⟷
((λ*x*::*real*. ((−*1*)^*n**(*n*+*1*)^(*n*+*1*)) $*_R$ (*ln x* ^ *n* * *x* ^ *n*)) *has-integral fact n*) {*0*<..<*1*}
proof (*rule has-integral-cong*)
fix *x* :: *real* **assume** *x*: $x \in$ {*0*<..<*1*}
have |*g′ x*| $*_R$ (*g x* ^ *n* / *exp* (*g x*)) =
(−*1*) ^ *n* * (*real n* + *1*) ^ (*n* + *1*) * *ln x* ^ *n* * (*exp* (*ln x* * (*n* + *1*)) / *x*)
using *x* **by** (*simp add*: *g-def g′-def exp-minus power-minus′ divide-simps add-ac*)
also have *exp* (*ln x* * (*n* + *1*)) = *x powr real* (*n* + *1*)
using *x* **by** (*simp add*: *powr-def*)
also have ... / *x* = *x* ^ *n*
using *x* **by** (*subst powr-realpow*) *auto*
finally show |*g′ x*| $*_R$ (*g x* ^ *n* / *exp* (*g x*)) =
((−*1*)^*n**(*n*+*1*)^(*n*+*1*)) $*_R$ (*ln x* ^ *n* * *x* ^ *n*)
by (*simp add*: *algebra-simps*)
qed
also have ... ⟷ ((λ*x*::*real*. *ln x* ^ *n* * *x* ^ *n*) *has-integral*
fact n $/_R$ *real-of-int* ((− *1*) ^ *n* * *int* ((*n* + *1*) ^ (*n* + *1*)))) {*0*<..<*1*}
by (*intro has-integral-cmul-iff′*) (*auto simp del*: *power-Suc*)
also have *fact n* $/_R$ *real-of-int* ((− *1*) ^ *n* * *int* ((*n* + *1*) ^ (*n* + *1*))) =
(−*1*) ^ *n* * *fact n* / (*n*+*1*) ^ (*n*+*1*)
by (*auto simp*: *divide-simps*)
finally show *?thesis*
by (*simp add*: *power-mult-distrib mult-ac*)
qed

1.4 Main proofs

We can now show the first formula: $\int_0^1 x^{-x}\,\mathrm{d}x = \sum_{n=1}^{\infty} n^{-n}$

lemma *sophomores-dream-aux1*:

```
  summable (λk. 1 / real ((k+1)^(k+1)))
  integral {0..1} (λx. x powr (−x)) = (∑n. 1 / (n+1)^(n+1))
proof −
  define S where S = (λx::real. ∑k. (−(x ∗ ln x)) ^ k / fact k)
  have S-eq: S x = x powr (−x) if x > 0 for x
  proof −
    have S x = exp (−x ∗ ln x)
      by (simp add: S-def exp-def field-simps)
    also have ... = x powr (−x)
      using ‹x > 0› by (simp add: powr-def)
    finally show ?thesis .
  qed

  have cont: continuous-on {0..1} (λx::real. ∑k<n. (−(x ∗ ln x)) ^ k / fact k) for n
    by (intro continuous-on-sum continuous-on-divide x-log-x-continuous continuous-on-power
           continuous-on-const continuous-on-minus) auto

  obtain I J where IJ: ⋀n. ((λx. ∑k<n. (−(x ∗ ln x)) ^ k / fact k) has-integral I n) {0..1}
                     (S has-integral J) {0..1} I ⟶ J
    using uniform-limit-integral [OF uniform-limit-sophomores-dream2 cont] by (auto simp: S-def)

  note ‹(S has-integral J) {0..1}›
  also have (S has-integral J) {0..1} ⟷ (S has-integral J) {0<..<1}
    by (simp add: has-integral-Icc-iff-Ioo)
  also have ... ⟷ ((λx. x powr (−x)) has-integral J) {0<..<1}
    by (intro has-integral-cong) (use S-eq in auto)
  also have ... ⟷ ((λx. x powr (−x)) has-integral J) {0..1}
    by (simp add: has-integral-Icc-iff-Ioo)
  finally have integral: ((λx. x powr (−x)) has-integral J) {0..1} .

  have I-eq: I = (λn. ∑k<n. 1 / real ((k+1)^(k+1)))
  proof
    fix n :: nat
    have ((λx::real. ∑k<n. (−1)^k ∗ ((x ∗ ln x) ^ k / fact k)) has-integral
          (∑k<n. (−1)^k ∗ ((−1)^k ∗ fact k / real ((k + 1) ^ (k + 1)) / fact k))) {0<..<1}
      by (intro has-integral-sum[OF - has-integral-mult-right] has-integral-divide
               sophomores-dream-aux-integral) auto
    also have (λx::real. ∑k<n. (−1)^k ∗ ((x ∗ ln x) ^ k / fact k)) =
              (λx::real. ∑k<n. (−(x ∗ ln x)) ^ k / fact k)
      by (simp add: power-minus')
    also have (∑k<n. (−1)^k ∗ ((−1) ^ k ∗ fact k / real ((k + 1) ^ (k + 1)) / fact k)) =
              (∑k<n. 1 / real ((k + 1) ^ (k + 1)))
      by simp
```

also note *has-integral-Icc-iff-Ioo* [*symmetric*]
finally show $I\ n = (\sum k{<}n.\ 1\ /\ real\ ((k{+}1)\hat{}(k{+}1)))$
by (*rule has-integral-unique* [*OF IJ*(*1*)[*of n*]])
qed
hence *sums*: $(\lambda k.\ 1\ /\ real\ ((k + 1)\ \hat{}\ (k + 1)))$ *sums J*
using *IJ*(*3*) *I-eq* **by** (*simp add*: *sums-def*)

from *sums* **show** *summable* $(\lambda k.\ 1\ /\ real\ ((k{+}1)\hat{}(k{+}1)))$
by (*simp add*: *sums-iff*)
from *integral sums* **show** *integral* $\{0..1\}\ (\lambda x.\ x\ powr\ (-x)) = (\sum n.\ 1\ /\ (n{+}1)\hat{}(n{+}1))$
by (*simp add*: *sums-iff has-integral-iff*)
qed

theorem *sophomores-dream1*:
$(\lambda k{::}nat.\ norm\ (k\ powi\ (-k)))$ *summable-on* $\{1..\}$
integral $\{0..1\}\ (\lambda x.\ x\ powr\ (-x)) = (\sum_\infty k{\in}\{(1{::}nat)..\}.\ k\ powi\ (-k))$
proof −
let *?I* = *integral* $\{0..1\}\ (\lambda x.\ x\ powr\ (-x))$
have $(\lambda k{::}nat.\ norm\ (k\ powi\ (-k)))$ *summable-on UNIV*
using *abs-summable-sophomores-dream*
by (*intro norm-summable-imp-summable-on*) (*auto simp*: *power-int-minus field-simps*)
thus $(\lambda k{::}nat.\ norm\ (k\ powi\ (-k)))$ *summable-on* $\{1..\}$
by (*rule summable-on-subset-banach*) *auto*

have $(\lambda n.\ 1\ /\ (n{+}1)\hat{}(n{+}1))$ *sums ?I*
using *sophomores-dream-aux1* **by** (*simp add*: *sums-iff*)
moreover have *summable* $(\lambda n.\ norm\ (1\ /\ real\ (Suc\ n\ \hat{}\ Suc\ n)))$
by (*subst summable-Suc-iff*) (*use abs-summable-sophomores-dream* **in** ‹*auto simp*: *field-simps*›)
ultimately have $((\lambda n{::}nat.\ 1\ /\ (n{+}1)\hat{}(n{+}1))$ *has-sum ?I*) *UNIV*
by (*intro norm-summable-imp-has-sum*) *auto*
also have *?this* $\longleftrightarrow$ $(((\lambda n{::}nat.\ 1\ /\ n\hat{}n) \circ Suc)$ *has-sum ?I*) *UNIV*
by (*simp add*: *o-def field-simps*)
also have … $\longleftrightarrow$ $((\lambda n{::}nat.\ 1\ /\ n\hat{}n)$ *has-sum ?I*) (*Suc* ‘ *UNIV*)
by (*intro has-sum-reindex* [*symmetric*]) *auto*
also have *Suc* ‘ *UNIV* = $\{1..\}$
using *greaterThan-0* **by** *auto*
also have $((\lambda n{::}nat.\ (1\ /\ real\ (n\ \hat{}\ n)))$ *has-sum ?I*) $\{1..\}$ $\longleftrightarrow$
$((\lambda n{::}nat.\ n\ powi\ (-n))$ *has-sum ?I*) $\{1..\}$
by (*intro has-sum-cong*) (*auto simp*: *power-int-minus field-simps power-minus'*)
finally show *integral* $\{0..1\}\ (\lambda x.\ x\ powr\ (-x)) = (\sum_\infty k{\in}\{(1{::}nat)..\}.\ k\ powi\ (-k))$
by (*auto dest*!: *infsumI simp*: *algebra-simps*)
qed

Next, we show the second formula: $\int_0^1 x^x\,\mathrm{d}x = -\sum_{n=1}^{\infty}(-n)^{-n}$

lemma *sophomores-dream-aux2*:
summable $(\lambda k.\ (-1)\ \hat{}\ k\ /\ real\ ((k{+}1)\hat{}(k{+}1)))$
integral $\{0..1\}\ (\lambda x.\ x\ powr\ x) = (\sum n.\ (-1)\hat{}n\ /\ (n{+}1)\hat{}(n{+}1))$

```
proof -
  define S where S = (λx::real. ∑k. (x * ln x) ^ k / fact k)
  have S-eq: S x = x powr x if x > 0 for x
  proof -
    have S x = exp (x * ln x)
      by (simp add: S-def exp-def field-simps)
    also have ... = x powr x
      using ‹x > 0› by (simp add: powr-def)
    finally show ?thesis .
  qed

  have cont: continuous-on {0..1} (λx::real. ∑k<n. (x * ln x) ^ k / fact k) for n
    by (intro continuous-on-sum continuous-on-divide x-log-x-continuous continuous-on-power
            continuous-on-const) auto

  obtain I J where IJ: ⋀n. ((λx. ∑k<n. (x * ln x) ^ k / fact k) has-integral I n) {0..1}
                     (S has-integral J) {0..1} I ⟶ J
    using uniform-limit-integral [OF uniform-limit-sophomores-dream1 cont] by (auto simp: S-def)

  note ‹(S has-integral J) {0..1}›
  also have (S has-integral J) {0..1} ⟷ (S has-integral J) {0<..<1}
    by (simp add: has-integral-Icc-iff-Ioo)
  also have ... ⟷ ((λx. x powr x) has-integral J) {0<..<1}
    by (intro has-integral-cong) (use S-eq in auto)
  also have ... ⟷ ((λx. x powr x) has-integral J) {0..1}
    by (simp add: has-integral-Icc-iff-Ioo)
  finally have integral: ((λx. x powr x) has-integral J) {0..1} .

  have I-eq: I = (λn. ∑k<n. (-1) ^ k / real ((k+1)^(k+1)))
  proof
    fix n :: nat
    have ((λx::real. ∑k<n. (x * ln x) ^ k / fact k) has-integral
          (∑k<n. (-1) ^ k * fact k / real ((k + 1) ^(k + 1)) / fact k)) {0<..<1}
      by (intro has-integral-sum has-integral-divide sophomores-dream-aux-integral) auto
    also have (∑k<n. (- 1) ^ k * fact k / real ((k + 1) ^(k + 1)) / fact k) =
               (∑k<n. (- 1) ^ k / real ((k + 1) ^(k + 1)))
      by simp
    also note has-integral-Icc-iff-Ioo [symmetric]
    finally show I n = (∑k<n. (-1) ^ k / real ((k+1)^(k+1)))
      by (rule has-integral-unique [OF IJ(1)[of n]])
  qed
  hence sums: (λk. (-1) ^ k / real ((k + 1) ^(k + 1))) sums J
    using IJ(3) I-eq by (simp add: sums-def)

  from sums show summable (λk. (-1) ^ k / real ((k+1)^(k+1)))
```

by (*simp add*: *sums-iff*)
from *integral sums* **show** *integral* $\{0..1\}$ $(\lambda x.\ x\ powr\ x) = (\sum n.\ (-1)\hat{}n\ /\ (n+1)\hat{}(n+1))$
by (*simp add*: *sums-iff has-integral-iff*)
qed

theorem *sophomores-dream2*:
$(\lambda k::nat.\ norm\ ((-k)\ powi\ (-k)))$ *summable-on* $\{1..\}$
integral $\{0..1\}$ $(\lambda x.\ x\ powr\ x) = -(\sum_\infty k\in\{(1::nat)..\}.\ (-k)\ powi\ (-k))$
proof −
let *?I* = *integral* $\{0..1\}$ $(\lambda x.\ x\ powr\ x)$
have $(\lambda k::nat.\ norm\ ((-k)\ powi\ (-k)))$ *summable-on UNIV*
using *abs-summable-sophomores-dream*
by (*intro norm-summable-imp-summable-on*) (*auto simp*: *power-int-minus field-simps*)
thus $(\lambda k::nat.\ norm\ ((-k)\ powi\ (-k)))$ *summable-on* $\{1..\}$
by (*rule summable-on-subset-banach*) *auto*

have $(\lambda n.\ (-1)\hat{}n\ /\ (n+1)\hat{}(n+1))$ *sums ?I*
using *sophomores-dream-aux2* **by** (*simp add*: *sums-iff*)
moreover have *summable* $(\lambda n.\ 1\ /\ real\ (Suc\ n\ \hat{}\ Suc\ n))$
by (*subst summable-Suc-iff*) (*use abs-summable-sophomores-dream* **in** ‹*auto simp*: *field-simps*›)
hence *summable* $(\lambda n.\ norm\ ((-\ 1)\ \hat{}\ n\ /\ real\ (Suc\ n\ \hat{}\ Suc\ n)))$
by *simp*
ultimately have $((\lambda n::nat.\ (-1)\hat{}n\ /\ (n+1)\hat{}(n+1))$ *has-sum ?I*) *UNIV*
by (*intro norm-summable-imp-has-sum*) *auto*
also have *?this* $\longleftrightarrow$ $(((\lambda n::nat.\ -((-1)\hat{}n\ /\ n\hat{}n)) \circ Suc)$ *has-sum ?I*) *UNIV*
by (*simp add*: *o-def field-simps*)
also have … $\longleftrightarrow$ $((\lambda n::nat.\ -((-1)\hat{}n\ /\ n\ \hat{}\ n))$ *has-sum ?I*) (*Suc* ‘ *UNIV*)
by (*intro has-sum-reindex* [*symmetric*]) *auto*
also have *Suc* ‘ *UNIV* = $\{1..\}$
using *greaterThan-0* **by** *auto*
also have $((\lambda n::nat.\ -((-\ 1)\ \hat{}\ n\ /\ real\ (n\ \hat{}\ n)))$ *has-sum ?I*) $\{1..\}$ $\longleftrightarrow$
$((\lambda n::nat.\ -((-n)\ powi\ (-n)))$ *has-sum ?I*) $\{1..\}$
by (*intro has-sum-cong*) (*auto simp*: *power-int-minus field-simps power-minus'*)
also have … $\longleftrightarrow$ $((\lambda n::nat.\ (-n)\ powi\ (-n))$ *has-sum* (−*?I*)) $\{1..\}$
by (*simp add*: *has-sum-uminus*)
finally show *integral* $\{0..1\}$ $(\lambda x.\ x\ powr\ x) = -(\sum_\infty k\in\{(1::nat)..\}.\ (-k)\ powi\ (-k))$
by (*auto dest*!: *infsumI simp*: *algebra-simps*)
qed

end

References

[1] J. Bernoulli. *Opera omnia*, volume 3. 1697.

[2] W. Dunham. *The Calculus Gallery: Masterpieces from Newton to Lebesgue*. Princeton University Press, 2004.

[3] Wikipedia contributors. Sophomore's dream — Wikipedia, the free encyclopedia. https://en.wikipedia.org/w/index.php?title=Sophomore%27s_dream&oldid=1053905038, 2021. [Online; accessed 10-April-2022].